
Einführung

Mit Access 95 hat Microsoft einen neuen Typ von Modulen eingeführt, die Klassenmodule. In Access 95 existierten sie nur als Formular- bzw. Berichtsmodule, seit Access 97 kann man Klassenmodule als eigenständige Module verwenden.

Klassenmodule unterscheiden sich von Standardmodulen, sie können verwendet werden, um eigene Objekte zu erstellen. Diese haben eigene Methoden und Ereignisse, können bei Bedarf instanziiert werden und es können mehrere gleichartige Objekte existieren.

Das erlaubt einen neuen Programmierstil, der nach dem objekt-orientierten Ansatz auch Objekt-basierte Programmierung genannt wird.

Mit Klassenmodulen können reale, sichtbare Objekte, wie z.B. Rechtecke ebenso modelliert werden wie „imaginäre“ Objekte, die nur innerhalb des Programms verwendet werden, z.B. Listen oder Stacks. Auch zur Kapselung komplizierterer Funktionen, etwa API-Aufrufe, sind sie gut geeignet.

Dieser Artikel soll eine Einführung in die Programmierung mit Klassenmodulen geben und anschließend anhand von praktischen Beispielen eine Reihe von Aspekten erläutern.

Eine weitere Neuerung in Access 97 ist das Schlüsselwort WithEvents, welches gemeinsam mit Klassenmodulen die Überlagerung von Ereignissen für bestehende Objekte erlaubt. Daraus ergeben sich sehr flexible und elegante Kombinationen, weshalb auch hierauf kurz eingegangen werden soll.

Programmierstile

Im wesentlichen werden die folgenden Programmierstile unterschieden:

· prozedural
Ursprünglicher und noch immer weit verbreiteter Ansatz, den die meisten Programmiersprachen unterstützen. Gefordert wird nur, daß die Sprache Techniken bereitstellt , die die Argument​übergabe an Funktionen und die Rücklieferung von Funktionswerten unterstützen. Aus der Sicht der Programmorganisation werden Funktionen dazu benutzt, Ordnung in ein Geflecht aus Algorithmen zu bringen.

Beipiel: sqrt()-Funktion (Quadratwurzel) in Fortran, Pascal, C

· modular
Verlegt den Schwerpunkt beim Programmentwurf vom Prozedur-Entwurf zur Organisation der Daten. Dabei werden aufeinander bezogene Prozeduren zusammen mit den Daten, die sie manipulieren, als Modul bezeichnet und soweit als möglich gekapselt („Geheimnisprinzip“). 
Beispiel: Stack-Modul, welches push() und pop() implementiert, z.B. in C, C++
· objektorientiert
Einführung von Objekten, die durch Identität, Zustand und Verhalten beschrieben werden. Diese Objekte unterstützen Abstraktion, Kapselung, Modularität und Vererbung. 

Beipiel: Stack-Klasse in C++
Im folgenden wird auf die Theorie objektorientierter Programmiersprachen eingegangen.

Was sind Objekte?

Objekte werden gekennzeichnet durch 

1. Abstraktion:
Hervorhebung charakteristischer Eigenschaften, Unterdrückung anderer Eigenschaften

Beipiel: Modellierung von geometrischen Formen: erst shape-Klasse mit Linienstärke, Farbe, Punkt; dann Klasse rectangle, circle, ... mit jeweils speziellen Zeichen- und Rechenfunktionen (Draw, Umfang, ...)

2. Kapselung:
Verbergen von Details, die nicht essentiell sind

Beispiel: interne Variablen, Implementierungsdetails

3. Modularität:
Zusammenfassung logischer Eigenschaften / Funktionen zu Objekten

zusätzlich bei Klassen (im Sinne von C++)

4. Vererbung:
Übernahme der Eigenschaften und Methoden der Basisklasse und Erweiterung dieser.

Dadurch entsteht die Möglichkeit einer hierarchischen  Gliederung von Typen, d.h. ein Subtyp erbt automatisch alle Eigenschaften seines Vaters, diese stehen auch allen Abkömmlingen zur Verfügung. Das führt zu kurzen, übersichtlichen Programmen, da die ganzen Basisfunktionen in Typbibliotheken ausgelagert werden können.

Beispiel: Basisklasse Shape habe nur die Eigenschaften Linienstärke, Füllfarbe und Nullpunkt. Die abgeleiteten Klassen Rectangle und Circle implementieren nur noch die Eigenschaften Breite/Höhe bzw. Durchmesser und dazu jeweils eine Methode Draw. Die Eigenschaften Linienstärke usw. stehen jetzt auch bei Benutzung der Klasse Rectangle zur Verfügung.

Objekte besitzen

· eine Identität, d.h. zwei Instanzen eines Objektes sind klar unterscheidbar, auch wenn sie identische Zustände besitzen

Beispiel: Klasse rectangle, zwei Instanzen r1 und r2 mit gleichem Zustand (gleichen Abmessungen usw.), trotzdem über r1 bzw. r2 getrennt ansprechbar

· einen Zustand, der durch die Belegung der privaten Variablen (Instanzvariablen) definiert ist

Beispiel: Klasse rectangle, r1 ist 2x3cm, r2 ist 3x4cm

· ein Verhalten, spezifiziert durch die Menge der Operationen (auf privaten Daten) bzw. durch den Aufruf anderer Objekte

Operationen:

· Modifikationen (ändern den Zustand)

· Selektionen (lesen den Zustand)

· Aktionen

· Konstruktoren (erzeugen Objekt, initialisieren den Zustand)

· Destruktoren (Freigabe von Speicher, löschen des Objekts)

In MS Access gibt es keine Vererbung, deshalb wird diese Art der Programmierung objektbasiert, nicht objektorientiert, genannt.

Objektbasiert – wozu ?

Im praktischen Einsatz bei der Programmierung in Visual Basic for Applications bietet die Verwendung von Objekten eine ganze Reihe von Vorteilen. Dazu gehören 

· Gruppierung logisch zusammengehörender Funktionen und Daten

· Übersichtlichkeit des Codes

· Unterstützung durch den Editor

· Wiederverwendbarkeit, auch in Bibliotheksdatenbanken (nur mit Tricks, s.u.)

· verschiedene Instanzen desselben Objektes mit unterschiedlichen Zuständen (private Daten) möglich 

Auf der anderen Seite steht ein gewisser Overhead beim Programmieren. 

Klassenmodule im Alltag

Erstellung

Klassenmodule zu Formularen und Berichten entstehen „einfach so“, wenn man eine Ereignisprozedur schreibt. Sie müssen nicht extra angelegt werden. Sind sie schon vorhanden, kann man sie (und den gesamten Code, den sie enthalten) über die Eigenschaft HasModule des Formulars oder Berichts löschen.

Formulare und Berichte sind auch nur Objekte. Über das zugehörige Klassenmodul kann man neue, selbstdefinierte Eigenschaften und Methoden anlegen (z.B. Form_BackColor) oder die Funktion bestehender verändern (Form_Open). Diese Objekte mit den neuen Methoden kann man genauso wie eigene Klassen verwenden.

Beispiel: öffnen von zwei Instanzen des gleichen Formulars

Dim f1 as New Form_Formular1

Dim f2 as New Form_Formular1

f1.BackColor = 0  ' diese Eigenschaft muß vorher selbst implementiert werden

f1.Visible = True

f2.Visible = True

Eigenständige Klassenmodule muß man selbst anlegen. Das funktioniert nicht mit dem „Neu“-Knopf im Reiter Module im Datenbankfenster, man muß dazu über das Menü Einfügen(Klassenmodul ein neues Modul einfügen. Dieses bekommt standardmäßig den Namen “Klasse1”. 

Dem Namen kommt eine besondere Bedeutung zu. Anders als bei Standardmodulen, bei denen der Name relativ unwichtig ist, wird der Name des Klassenmoduls der Name des Objekts, welcher bei der Deklaration verwendet wird. Nur wenn das Klassenmodul Rectangle heißt, kann man später ein Objekt r1 mit der Deklaration

Dim r1 as New Rectangle

erzeugen. Hier spielt unschönerweise die Benennung in die Funktion herein, sinnvoller wäre ein eigenes Statement der Art

ClassModuleName = "Rectangle"

' gibt es nicht

im Deklarationsteil des Klassenmoduls. Gibt es aber nicht.

Bestehende Klassenmodule kann man über die Zwischenablage oder über Datei(Externe Daten(Importieren in eine andere Datenbank kopieren. 

Ein als Text gespeichertes Klassenmodul muß man über Einfügen(Datei in ein neu erzeugtes Klassenmodul einfügen, das Einfügen in ein Standardmodul funktioniert nicht.

Unterschiede zu Standardmodulen

Wie oben im theoretischen Teil schon kurz erläutert besitzen Objekte ein Verhalten, welches durch Operationen (Konstruktoren, Destruktoren, Modifikationen, Selektionen, Aktionen) be​schrieben wird. Bei Klassenmodulen sehen diese Operationen folgendermaßen aus:

Konstruktor, Destruktor (Class_Initialize, Class_Terminate)

Jede Klasse kann zwei spezielle Prozeduren Class_Initialize() und Class_Terminate() besitzen. 

Beim Erzeugen des Objektes wird der Code in Class_Initialize ausgeführt. Diese Stelle bietet sich an, um interne Variablen mit vernünftigen Voreinstellung zu belegen oder benötigte Objekte zu initialisieren.

Bevor das Objekt wieder gelöscht wird (entweder durch Zuweisung von Nothing an die Objektvariable oder durch Verlassen des Gültigkeitsbereiches) wird der Code in Class_Terminate ausgeführt. Hier können intern verwendete Objekte wieder freigegeben oder sonstige abschließende Arbeiten ausgeführt werden.

Keine dieser beiden Prozeduren ist von außen sichtbar, wenn sie vorhanden sind werden sie automatisch aufgerufen.

Beispiel: zeigt den Anfang einer möglichen Erweiterung des Collection-Objektes. 

Collection ist ein von MS vordefiniertes Objekt, daß Einträge beliebiger Typen aufnimmt, durch die man dann mit For...Each oder For...Next Schleifen durchgehen kann. Zusätzlich kann man beim hinzufügen eines neuen Elements noch einen Index angeben, der eindeutig sein muß. Damit läßt sich nebenbei und völlig ohne Anstrengung noch eine Duplikateliminierung erreichen. Der Index kann ebenfalls von einem beliebigen Datentyp sein, über ihn kann später auf das Element zugegriffen werden. Das ermöglicht sogenannte „assoziative Arrays“, wie sie in Scriptsprachen wie AWK und Perl verwendet werden. 

Beispiel: c sei eine Collection mit Gehalt, die Position der Angestellten sei der Index. Dann kann man nach c("Chef") oder c("Putzfrau") fragen und kriegt das jeweilige Gehalt zurück. Mehr dazu in der Online-Hilfe.

Zusätzlich könnte z.B. eine Methode „Sort“ implementiert werden, die alle Einträge sortiert.

Private c As Collection

Private Sub Class_Initialize()

    Set c = New Collection

End Sub

Private Sub Class_Terminate()

    Set c = Nothing

End Sub 

Modifikationen (Property Let, Property Set)

Access kennt 2 sehr ähnliche Typen von Modifikationen, Property Let und Property Set. Mit Let erfolgt die Zuweisung einfacher Werte (String, Zahl, ...), bei Set wird ein Objekt an die Klasse übergeben. 

Selektionen (Property Get)

Mit Property Get werden Werte von einer Klasse zurückgegeben. 

Diese 3 Property-Funktionen bilden zusammen die Eigenschaften des Objektes. Über sie können die internen Variablen belegt werden, die das Verhalten beim Aufruf einer Aktion (Methode) bestimmen.

Beispiel: zeigt die einfache Implementierung einer Klasse für Rechtecke, bei der Breite und Höhe angegeben werden muß und anschließend der Umfang abgefragt werden kann. Dieser Code sei in einem Klassenmodul namens clsRectangle gespeichert.

Private prvH As Double

Private prvB As Double

Property Let Breite(b As Double)

    prvB = b

End Property

Property Let Hoehe(h As Double)

    prvH = h

End Property

Property Get Umfang() As Double

    Umfang = 2 * prvH + 2 * prvB

End Property

Diese Klasse kann dann folgendermaßen benutzt werden, der Aufruf erfolgt in einem beliebigen Modul (auch ein anderes Klassenmodul):

Dim r1 As New clsRectangle

r1.Breite = 2

r1.Hoehe = 3

Debug.Print r1.Umfang

Aktionen (Methoden)

Aktionen sind ganz normale Funktionen (Function) oder Prozeduren (Sub), die sich nicht von denen in Standardmodulen unterscheiden. Der Name der Funktion wird der Name der Methode des Objektes. Als Private deklarierte Funktionen werden nicht nach außen sichtbar gemacht und können demzufolge auch nicht als Methode aufgerufen werden.

Beispiel: hinzufügen der Methoden AddItem und RemoveItem zu der Collection-Klasse aus dem vorletzten Beispiel, RemoveItem liefert False, wenn beim Entfernen etwas schiefgegangen ist, sonst True. Die Fehlerbehandlung ist verbesserungbedürftig, um das grundlegende Vorgehen zu zeigen reicht es aber allemal.

Sub AddItem(s As String)

    c.Add s

End Sub

Function RemoveItem(idx As Integer) As Boolean

On Error GoTo Fehler

    c.Remove idx

    RemoveItem = True

    Exit Function

Fehler:

    Err.Clear

    RemoveItem = False

End Function

Lebensdauer

Wie bei allen anderen Variablen beginnt die Lebensdauer einer Objektvariablen (und damit u.U. auch des zugehörigen Objektes, siehe die Fälle a] und b]) mit der Deklaration und endet mit dem Verlassen des Gültigkeitsbereiches. Daraus folgt, daß man Objekte, die man im ganzen Formular verwenden will, im Deklarationsteil des Formular-Klassenmoduls deklarieren muß. Für eine Initialisierung (Fall b]) bietet sich wahlweise die Ereignisse OnOpen bzw. OnLoad an. 

In Access sind die Lebensdauer einer (Objekt-)Variablen und die Lebensdauer des zugehörigen Objekts eng gekoppelt. Es gibt 2 Varianten der Erzeugung eines neuen Objektes.

a] Erzeugung des Objekts bei der Deklaration der Variablen:

Dim r1 as New Rectangle

b] Deklaration der Variablen, Erzeugen des Objekts zu einem späteren Zeitpunkt:

Dim r1 as Rectangle

Deklaration der Variablen
Set r1 = New Rectangle

Erzeugen des Objektes
Die Löschung des Objektes kann entweder implizit erfolgen, wenn der Gültigkeitsbereich der Variablen verlassen wird (z.B. Ende der Funktion, Schließen des Formulars) oder es wird explizit mit 

Set fd = Nothing

gelöscht. Die Form ist vorzuziehen, dabei wird ein Bug in Access 97 umgangen. Der führt manchmal dazu, daß eine Objektvariable (der Verweis auf das Objekt) zwar gelöscht, der Speicherplatz aber nicht freigegeben wird. Danach läßt sich Access nicht mehr beenden und verschwindet nur in die Task​leiste (siehe FAQ).

WithEvents - Einklinken in die Ereignisausführung

Ein weiteres neues Sprachelement, welches mit Access 97 eingeführt wurde und im Zusammenhang mit Klassenmodulen interessant ist, heißt WithEvents. Damit lassen sich „Templates“ (Vorlagen) zu Ereignisprozeduren erstellen, welche dann für jedes Objekt dieses Typs beim Auslösen des jeweiligen Ereignisses ausgeführt werden. 

Beispielsweise kann man ein Template für Textfelder erstellen, die beim Hineingehen die Hintergrundfarbe ändern. Denkbar sind auch Buttons, die in vielen verschiedenen Formularen die gleiche Funktion haben. Der Code dafür muß nur einmal vorhanden sein, was die Pflege erleichtert.

Funktion

Mit WithEvents können nur die vordefinierten Ereignisse von eingebauten Objekten überlagert werden. Es ist also nicht möglich, neue Ereignisse zu definieren.

Befindet sich in der jeweiligen Ereignisprozedur des Objektes noch Code, so wird dieser zuerst ausgeführt. Erst danach werden die im Klassenmodul festgelegten Aktionen ausgelöst. Diese Reihenfolge ist meiner Meinung nach ungeschickt, man könnte sonst im Klassenmodul allgemeine Einstellungen setzen und danach die für das jeweilige Objekt spezifischen Einstellungen, das ist leider nicht möglich.

Anwendung, Besonderheiten

Das Schlüsselwort WithEvents kann nur in Klassenmodulen verwendet werden. Zum Überlagern eines Ereignisses sind folgende Schritte notwendig (am Beispiel eines Textfeldes, daß beim Hineingehen seine Hintergrundfarbe verändert und beim Herausgehen die alte Farbe wieder herstellt):

Im Klassenmodul wird eine Eigenschaft programmiert, über die die Zuordnung eines bestehenden Textfeldes (im Formular) zur neuen Klasse hergestellt wird. Dann werden die zu überlagernden Eigenschaften programmiert:

1. Erstellen eines Klassenmoduls clsTextbox, welches eine private Variable vom Typ Textbox enthält, die mit dem WithEvents-Schlüsselwort deklariert wird:

Private WithEvents m_Textbox As TextBox

2. Erstellen einer Property Set – Funktion, über die der Variablen m_Textbox ein Textfeld in einem Formular zugewiesen wird, für das die Eigenschaften geändert werden sollen:

Public Property Set theTextbox(box As TextBox)

    Set m_Textbox = box

End Property

3. Programmieren der Ereignisse, die überlagert werden sollen:

Private Sub m_Textbox_Enter()

    SaveHintergrund = m_Textbox.BackColor

    m_Textbox.BackColor = 8454143 'gelb

End Sub

Private Sub m_Textbox_Exit(Cancel As Integer)

    m_Textbox.BackColor = SaveHintergrund

End Sub

Im Formular muß für jedes Textfeld, dessen Eigenschaften verändert werden sollen, eine Variable angelegt werden. Nur über diese kann die Zuordnung zu der neuen Klasse hergestellt werden.

4. Deklarieren einer Variablen myText1 vom Typ clsTextbox im Deklarationsteil des Formularmoduls:

Dim myText1 As New clsTextbox

5. Zuweisen des bereits bestehenden Textfeldes an die Variable mit Hilfe der oben definierten Eigenschaft der neuen Klasse beim öffnen des Formulars

Private Sub Form_Load()

    Set myText1.theTextbox = Me!Text1

End Sub

Damit sollten die neu definierten Ereignis-Eigenschaften für das Textfeld gelten, da alle notwendigen Zuweisungen erfolgt sind. Leider ist dem nicht so, da offenbar das Ereignis nicht ausgelöst wird, wenn in der zugehörigen Eigenschaft nicht [Ereignisprozedur] steht. Deswegen

6. Eintragen von [Ereignisprozedur] bei den Ereignissen OnEnter und OnExit des Textfeldes. Das ist alles, es ist nicht notwendig, irgendwelchen Code einzugeben. Aus Gründen der Nachvollziehbarkeit sollte man aber wenigstens eine Zeile Kommentar schreiben.

Zusammenfassung

Hier noch einmal das ganze Verfahren im Zusammenhang.

Verwendung in Bibliotheks-Datenbanken

Bei größeren Projekten, die modular aufgebaut sind, bietet sich die Verwendung von Bibliotheks-Datenbanken an. Dort werden Code und Objekte abgelegt, die von mehreren Projektteilen aufgerufen werden können. Auf diese Weise steht der Code nur an einer einzigen Stelle und muß nur einmal gepflegt werden.

Auch Klassenmodule können in Bibliotheksdatenbanken ausgelagert werden. Dann „sieht“ allerdings weder Access 97 noch Access 2000 diese Klassenmodule, auch wenn ein korrekter Verweis auf die Bibliotheks-Datenbank besteht.

Das folgende ist eine undokumentierte Lösung dieses Problems.

1. Selektieren Sie im Datenbankfenster das Klassenmodul, exportieren Sie es als Textdatei (Verwenden Sie nicht „Als Text speichern“ in der Entwurfsansicht)

2. Öffnen Sie die Textdatei in einem Texteditor (z.B. Notepad). Die ersten vier Zeilen enthalten Attribute, eine davon heißt 

Attribute VB_Exposed = False

3. Ändern Sie diese Zeile in 

Attribute VB_Exposed = True
4. Löschen Sie das Klassenmodul aus Ihrer Bibliotheks-Datenbank (sofern schon dort vorhanden). Erstellen Sie ein neues, leeres Klassenmodul (Einfügen(Klassenmodul).

5. Wählen Sie Einfügen(Datei und importieren Sie die soeben bearbeitete Textdatei. Speichern Sie das Klassenmodul unter dem Namen, den es vorher auch hatte.

Damit ist diese Klasse nach außen sichtbar, sofern ein Verweis auf die Bibliotheks-Datenbank besteht.

Beispiele

Die Klasse Sort

Nicht immer muß ein Klassenmodul ein Objekt modellieren. Auch zur Kapselung von ganz normalen wiederverwendbaren Funktionen eignet es sich gut. In diesem Beispiel soll eine Klasse vorgestellt werden, die nur eine Methode hat (.Sort), welche ein Array aufsteigend sortiert.

Der ganze Code könnte unverändert auch in einem normalen Modul stehen. Die Verwendung eines Klassenmoduls bringt aber den Vorteil, daß schon bei der Eingabe im Editor die Methode sichtbar ist und so Schreibfehler vermieden werden können. Außerdem könnte man mehrere Sortierverfahren für verschiedene Zwecke implementieren, die dann schon bei der Programmierung aufgelistetet werden. Weiterhin lassen sich bei der Initialisierung und der Terminierung Aktionen ausführen, was in einem normalen Modul nicht möglich ist.

Wie schon gesagt besitzt die Klasse Sort nur eine Methode und keinerlei interne Variablen. Beim Aufruf dieser Methode wird die interne Funktion QuickSortArray (QSArray) aufgerufen, welche das übergebene Array vom Typ Variant mit dem Quicksort-Algorithmus aufsteigend sortiert. Dabei wird „am Platz“ sortiert, d.h. die Methode verändert das übergebene Array. Die Arraygröße ist dabei beliebig.

Public Sub Sort(arrIn() As Variant)

QSArray arrIn, LBound(arrIn()), UBound(arrIn())

End Sub

Private Sub QSArray(arrIn() As Variant, _

                    ByVal intLowBound As Integer, _

                    ByVal intHighBound As Integer)

  'Implementierung von QSArray

End Sub

Die weitere Arbeit übernimmt die Funktion QSArray(), die sich selbst rekursiv aufruft.

Diese Klasse läßt sich beliebig erweitern. Mögliche Verbesserungen wären ein zweiter Sortieralgorithmus (z.B. BubbleSort) und der Aufruf eines der beiden Algorithmen abhängig von der Anzahl der zu sortierenden Elemente (oder anderer Kriterien). Ebenfalls denkbar ist die Einführung einer neuen Eigenschaft SortOrder, über die entschieden wird, ob auf- oder absteigend sortiert werden soll.

Die Klasse FileDialog

Die Klasse FileDialog ist ein Beispiel für die Kapselung von API-Aufrufen in einem Klassenmodul. Verpackt werden hier die Aufrufe zur Dateiauswahl und zur Auswahl eines Verzeichnisses.

Als API-Aufrufe werden die Aufrufe von Funktionen bezeichnet, die in DLLs zur Verfügung stehen und von allen Programmen genutzt werden können. Diese Funktionen sind eigentlich für den Aufruf aus Programmiersprachen gedacht, die in Bezug auf Adressierung und Speicherverwaltung sehr viel flexibler sind als VBA (z.B. C, C++). Trotzdem können sie aus VBA aufgerufen werden, allerdings ist ihre Verwendung dann mit diversen Tücken behaftet und erfordert einiges Wissen über Speicherverwaltung, Pointer, zu setzende Parameter usw.

Diese Komplexität kann man gut in einem Klassenmodul verstecken. Dem Benutzer werden nur Eigenschaften und Methoden präsentiert, die verständlich und einfach zu verwenden sind. Eine solche Klasse kann sehr fehlertolerant sein, indem z.B. alle wichtigen Parameter bei der Erzeugung des Objektes (Class_Inintialize, siehe Seite 7) gesetzt werden. Außerdem kann man APIs aus unterschiedlichen DLLs zusammen verwenden, so daß in der Klasse ähnliche Funktionen zusammengefaßt werden können, ohne daß der Benutzer sich für die Hintergründe interessieren muß.

Ein solches Beispiel ist die Klasse FileDialog, die das Standard-Windows-Fenster zum Öffnen und Speichern einer Datei anzeigt und die ausgewählte Datei zurückliefert. Diese Fenster werden (zusammen mit der Auswahl von Farbe) als Common Dialog bezeichnet, da sie von allen Windwos-Programmen benutzt werden (können), die entsprechende Bibliothek ist die comdlg32.dll.

Verwandt dazu ist die Auswahl (nur) eines Verzeichnisses, die mit diesen Dialogen nicht möglich ist. Dazu wird ein API-Aufruf  aus der Bibliothek shell32.dll benutzt.

Durch die Kapselung reduziert sich der Aufwand beim Aufruf auf wenige Zeilen, die allereinfachste Version ist 

Dim fd As New FileDialog

Dim Dateiname as String

Dateiname = fd.ShowOpen           ' oder .ShowSave

if Dateiname = "" then exit sub   ' Abbruch durch Benutzer

Zusätzlich können spezifische Einstellung vorgenommen werden, etwa durch Setzen des Dialog-Titels, Setzen von Dateifiltern und ähnliches. Eine Erläuterung aller möglichen Parameter würde hier zu weit führen. Ein Demoformular befindet sich in der beiliegenden Datenbank.

Dim myText1 As New clsTextbox


Set myText1.theTextbox = Me!Text1





Ein Textfeld im Formular wird einer Variablen vom Typ clsTextbox zugewiesen.





Im Deklarationsteil von clsTextbox wird eine private Variable vom Typ Textbox mit dem Schlüsselwort With�Events deklariert.





Private WithEvents m_Textbox As TextBox





Private Sub m_Textbox_Enter()


Private Sub m_Textbox_Exit(Cancel _


As Integer)





Im Klassenmodul werden die notwendigen Ereignisse für diese Variable programmiert





In die Eigenschaften des Textfeldes im Formular wird [Ereignisprozedur] eingetragen.

ClsTextbox ist ein selbst erstelltes Klassenmodul,

